Computation of Asymptotic Least Squares Estimates

Söderström, Torsten

1973

Document Version:
Publisher's PDF, also known as Version of record

Link to publication

Citation for published version (APA):
Söderström, T. (1973). *Computation of Asymptotic Least Squares Estimates.* (Technical Reports TFRT-7051). Department of Automatic Control, Lund Institute of Technology (LTH).

Total number of authors:
1

COMPUTATION OF ASYMPTOTIC LEAST SQUARES ESTIMATES

T. SÖDERSTRÖM

Report 7314 (C) July 1973
Lund Institute of Technology
Division of Automatic Control

COMPUTATION OF ASYMPTOTIC LEAST SQUARES ESTIMATES.

T. Söderström

ABSTRACT.

In this report programs are given which computes asymptotic least squares estimates.

TABLE OF CONTENTS

I. DESCRIPTION OF THE PROGRAMS.

Section II of the report contains complete lists of the programs. A short description of the programs are given below. The programs not described (CINT, ORDER and SYMIN) can be found in the program library of the Division of Automatic Control.

1. TLST. This is a main (test) program. Parameters are read and written and a CALL LST is done. It is possible to do repeated calls of LST in one execution.

2. LST. It is assumed that the system

$$A(q^{-1})y(t) = B(q^{-1})u(t) + C(q^{-1})e(t) \qquad (1)$$

is given where

$$A(q^{-1}) = 1 + a_1q^{-1} + \dots + a_{NSA}q^{-NSA}$$

$$B(q^{-1}) = b_1q^{-1} + \dots + b_{NSB}q^{-NSB}$$

$$C(q^{-1}) = 1 + c_1q^{-1} + \dots + c_{NSC}q^{-NSC}$$

$u(t)$ white noise $Eu(t) = 0 \quad Eu(t)^2 = VU$
$e(t)$ white noise $Ee(t) = 0 \quad Ee(t)^2 = VE$
$u(t)$ and $e(t)$ independent.

An LS identification of the system is simulated in the sense that the asymptotic values of the estimates are computed using complex integrals.

The following model is obtained

$$\hat{A}(q^{-1})y(t) = \hat{B}(q^{-1})u(t) + \varepsilon(t)$$

where

$$\hat{A}(q^{-1}) = 1 + \hat{a}_1 q^{-1} + \ldots + \hat{a}_{NMA} q^{-NMA}$$

$$\hat{B}(q^{-1}) = \hat{b}_1 q^{-1} + \ldots + \hat{b}_{NMB} q^{-NMB}$$

The following variables are obtained

$TM = [\hat{a}_1 \ \ldots \ \hat{a}_{NMA} \ \hat{b}_1 \ \ldots \ \hat{b}_{NMB}]$

V - the loss function $= \Sigma\varepsilon^2(t)$

S - estimated standard deviation of $\varepsilon(t)$

P - estimated covariance matrix of TM

CC - estimated standard deviation of the elements of TM.

3. <u>COVM</u>. It is assumed that the system (1) is given and that N = NSA = NSB = NSC. The following matrix is computed.

$$COV = \begin{pmatrix}
r_y(0) & r_y(1) & r_y(NC-1) & r_{yu}(0) & & r_{yu}(NC-1) & r_{ye}(0) & & r_{ye}(NC-1) \\
\vdots & \ddots & \vdots & \vdots & \ddots & \vdots & \vdots & \ddots & \vdots \\
r_y(NC-1) & & r_y(0) & r_{yu}(1-NC) & & r_{yu}(0) & r_{ye}(1-NC) & & r_{ye}(0) \\
r_{yu}(0) & & r_{yu}(1-NC) & r_u(0) & & r_u(NC-1) & r_{ue}(0) & & r_{ue}(NC-1) \\
\vdots & \ddots & \vdots & \vdots & \ddots & \vdots & \vdots & \ddots & \vdots \\
r_{yu}(NC-1) & & r_{yu}(0) & r_u(NC-1) & & r_u(0) & r_{ue}(1-NC) & & r_{ue}(0) \\
r_{ye}(0) & & r_{ye}(1-NC) & r_{ue}(0) & & r_{ue}(1-NC) & r_e(0) & & r_e(NC-1) \\
\vdots & \ddots & \vdots & \vdots & \ddots & \vdots & \vdots & \ddots & \vdots \\
r_{ye}(NC-1) & & r_{ye}(0) & r_{ue}(NC-1) & & r_{ue}(0) & r_e(NC-1) & & r_e(0)
\end{pmatrix}$$

COV is the covariance matrix of the vector

[y(t-1) ... y(t-NC) u(t-1) ... u(t-NC) e(t-1) ... e(t-NC)]

5. PMPY. This program performs polynomial multiplication, see the comments in the list.

```
C     PROGRAM TLST
C
C     TEST OF SUBROUTINE LST
C     AUTHOR TORSTEN SODERSTROM 1971-10-05
C
C     SUBROUTINE REQUIRED
C             LST
C             COVM
C             CINT
C             ORDER
C             PMPY
C             GJRV
C
      DIMENSION TS(30),TM(20),P(20,20),C(20)
C
    1 READ 101,M,NA,NB,NC
  101 FORMAT(10I5)
      IF(M.EQ.0) GO TO 99
      N=NA+NB+NC
      IF(N.GT.0) READ 100,(TS(I),I=1,N),VU,VE
      IF(N.EQ.0) READ 100,VU,VE
  100 FORMAT(8F10.5)
    2 READ 101,NMA,NMB
      IF(NMA*NMB.EQ.0) GO TO 1
C
      PRINT 200
  200 FORMAT(1H1,09X,'SYSTEM'/10X,6(1H*))
      PRINT 201,NA,NB,NC,M
  201 FORMAT(/10X,'NUMBER OF A-PARAMETERS',I5
     */10X,'NUMBER OF B-PARAMETERS',I5
     */10X,'NUMBER OF C-PARAMETERS',I5
     */10X,'NUMBER OF SAMPLING INTERVALS',I8)

      PRINT 202
  202 FORMAT(/10X,'PARAMETERS')
      IF(N.GT.0) PRINT 100,(TS(I),I=1,N)
      PRINT 204,VU,VE
  204 FORMAT(/10X,'VARIANCE OF THE INPUT SIGNAL',G12.5
     */10X,'VARIANCE OF THE NOISE',7X,G12.5)
      PRINT 205
  205 FORMAT(///10X,'MODEL'/10X,5(1H*))
      PRINT 206,NMA,NMB
  206 FORMAT(/10X,'NUMBER OF A-PARAMETERS',I5
     */10X,'NUMBER OF B-PARAMETERS',I5)
C
      CALL LST(TS,VU,VE,TM,V,S,P,C,NA,NB,NC,NMA,NMB,M,1,20)
C
      GO TO 2
C
   99 CONTINUE
      STOP
      END
```

```
      SUBROUTINE LST(TS,VU,VE,TM,V,S,P,CC,NSA,NSB,NSC,NMA,NMB,M,IPR,IA)
C
C     GIVEN THE SYSTEM
C     A(Q)*Y(T)=B(Q)*U(T)+C(Q)*E(T)
C     A(Q)=1+A(1)*Q**(-1)+...+A(NSA)*Q**(-NSA)
C     B(Q)=  B(1)*Q**(-1)+...+B(NSB)*Q**(-NSB)
C     C(Q)=1+C(1)*Q**(-1)+...+C(NSC)*Q**(-NSC)
C     U(T) AND E(T) INDEPENDENT WHITE NOISES WITH VARIANCES VU RESP VE
C
C     COMPUTED THE THEORETICAL, ASYMPTOTIC LEAST SQUARES MODEL
C     AM(Q)*Y(T) = BM(Q)*U(T) + EPS(T)
C     AND RELATED VARIABLES
C     AM(Q)=1+AM(1)*Q**(-1)+...+AM(MNA)*Q**(-MNA)
C     BM(Q)=  BM(1)*Q**(-1)+...+BM(MNB)*Q**(-MNB)
C
C     AUTHOR TORSTEN SODERSTROM 1971-10-05
C     REFERENCE REPORT 7314(C)
C
C     TS-VECTOR OF ORDER (NSA+NSB+NSC)
C     TS=(A(1),...A(NSA),B(1),...B(NSB),C(1),...C(NSC))
C     VU-VARIANCE OF U(T)
C     VE-VARIANCE OF E(T)
C     TM-VECTOR OF ORDER (NMA+NMB)
C     TM=(AM(1),...AM(NMA),BM(1),...BM(NMB))
C     V - THE LOSS FUNCTION (=SUM OF SQUARED RESIDUALS)
C     S-ESTIMATED STANDARD DEVIATION OF THE NOISE
C     P-ESTIMATED COVARIANCE MATRIX OF TM
C     CC-ESTIMATED STANDARD DEVIATIONS OF THE PARAMETERS
C     NSA,NSB,NSC -ORDERS OF A,B,C (MIN0,MAX10)
C     NMA,NMB-ORDERS OF AM,BM (MIN 0,MAX 9)
C     M-NUMBER OF SAMPLES
C     IPR-PRINT PARAMETER
C     IPR=0 NOTHING PRINTED
C         1 OUTPUT GENERATED (T,CC,P,V,S ARE PRINTED)
C     IA-DIMENSION PARAMETER OF P
C
C     SUBROUTINE REQUIRED
C            COVM
C            CINT
C            ORDER
C            PMPY
C            SYMIN
C
      DIMENSION TS(1),TM(1),P(IA,IA),CC(1)
      DIMENSION COV(30,30),R(20),A(10),B(10),C(10)
C
      N=MAX0(NSA,NSB,NSC)
      NN=NMA+NMB
      IF(N*NN.EQ.0) GO TO 99
      NC=MAX0(NMA,NMB)+1
C
      DO 1 I=1,N
      A(I)=0.
      B(I)=0.
    1 C(I)=0.
      IF(NSA.EQ.0) GO TO 3
      DO 2 I=1,NSA
    2 A(I)=TS(I)
    3 IF(NSB.EQ.0) GO TO 5
      DO 4 I=1,NSB
    4 B(I)=TS(I+NSA)
    5 IF(NSC.EQ.0) GO TO 7
      DO 6 I=1,NSC
```

```
    6 C(I)=TS(I+NSA+NSB)
C
    7 CALL COVM(A,B,C,COV,VU,VE,N,NC,30)
C
C     COMPUTE P=LIM FI(T)*FI/M AND R=LIM FI(T)*Y/M
C
      IF(NMA.EQ.0) GO TO 11
      DO 10 I=1,NMA
      DO 8 J=1,NMA
    8 P(I,J)=COV(I,J)
      R(I)=-COV(1,I+1)
      IF(NMB.EQ.0) GO TO 10
      DO 9 J=1,NMB
    9 P(I,J+NMA)=-COV(I,J+NC)
   10 CONTINUE
   11 IF(NMB.EQ.0) GO TO 13
      DO 12 I=1,NMB
      R(I+NMA)=COV(1,1+NC+I)
      DO 12 J=1,NMB
   12 P(I+NMA,J+NMA)=COV(I+NC,J+NC)
      DO 14 I=1,NN
      DO 14 J=1,I
   14 P(I,J)=P(J,I)
   13 CONTINUE
C
C     COMPUTE THE SOLUTION
C
      CALL SYMIN(NN,IA,IFAIL,P)
      IF(IFAIL.EQ.0) GO TO 31
      PRINT 116
  116 FORMAT(/10X,'THE SYSTEM IS SINGULAR')
      GO TO 99
   31 CONTINUE
C
      DO 15 I=1,NN
      TM(I)=0.
      DO 15 J=1,NN
   15 TM(I)=TM(I)+P(I,J)*R(J)
      V=COV(1,1)
      DO 16 I=1,NN
   16 V=V-R(I)*TM(I)
      IF(V.LT.0.0) S=SQRT(-V)
      IF(V.GE.0.0) S=SQRT(V)
      V=V*M
      DO 17 I=1,NN
      DO 17 J=1,NN
   17 P(I,J)=P(I,J)*S*S/M
      DO 18 I=1,NN
      IF(P(I,I).LT.0.0) CC(I)=SQRT(-P(I,I))
   18 IF(P(I,I).GE.0.0) CC(I)=SQRT( P(I,I))
C
C     PRINT RESULTS
C
      IF (IPR) 99,99,19
   19 PRINT 101
  101 FORMAT(/10X,'THEORETICAL LS ESTIMATE'/10X,23(1H-)
     *//10X,'PARAMETER ESTIMATES')
      PRINT 100,(TM(I),I=1,NN)
  100 FORMAT(10G12.5)
      PRINT 102
  102 FORMAT(/10X,'STANDARD DEVIATIONS')
      PRINT 100,(CC(I),I=1,NN)
      PRINT 103
  103 FORMAT(/10X,'COVARIANCE MATRIX')
```

```
      DO 20 I=1,NN
   20 PRINT 100,(P(I,J),J=1,NN)
      PRINT 104
  104 FORMAT(/10X,'LOSS FUNCTION')
      PRINT 100,V
      PRINT 105
  105 FORMAT(/10X,'STANDARD DEVIATION OF THE NOISE')
      PRINT 100,S
C
   99 CONTINUE
      RETURN
      END
```

```
      SUBROUTINE COVM(A,B,C,COV,VU,VE,N,NC,IA)
C
C     GIVEN THE SYSTEM
C     A(Q)*Y(T)=B(Q)*U(T)+C(Q)*E(T)
C
C     A(Q)=1+A(1)*Q**(-1)+...A(N)*Q**(-N)
C     B(Q)=  B(1)*Q**(-1)+...+B(N)*Q**(-N)
C     C(Q)=1+C(1)*Q**(-1)+...+C(N)*Q**(-N)
C
C     U(T) WHITE NOISE
C     E(T) WHITE NOISE
C     U(T) AND E(T) INDEPENDENT
C
C     COMPUTED THE EXTENDED SYSTEM COVARIANCE MATRIX
C            RY   RYU  RYE
C     COV = (RUY  RU   RUE )
C            REY  REU  RE
C
C       =COVARIANCE MATRIX OF THE VECTOR
C       (Y(T-1) Y(T-2)...Y(T-NC) U(T-1)...U(T-NC) E(T-1)...E(T-NC) )
C
C     AUTHOR TORSTEN SODERSTROM 1971-12-24
C     REFERENCE REPORT 7314(C)
C
C     A VECTOR OF ORDER N
C     B VECTOR OF ORDER N
C     C VECTOR OF ORDER N
C     COV-MATRIX OF ORDER (3*NC)*(3*NC), AT RETURN CONTAINING THE
C     EXTENDED SYSTEM COVARIANCE MATRIX
C     VU-VARIANCE OF U(T)
C     VE-VARIANCE OF E(T)
C     N-ORDER OF THE SYSTEM (MIN1,MAX10)
C     3*NC-ORDER IF THE MATRIX COV
C     NC (MIN 1,MAX 10)
C     IA-DIMENSION PARAMETER OF COV
C
C     SUBROUTINE REQUIRED
C            CINT
C            ORDER
C            PMPY
C
      DIMENSION A(1),B(1),C(1),COV(IA,1)
      DIMENSION RYE(20),RYU(20),RYYE(10),RYYU(10)
      DIMENSION AI(20),BI(20),CIU(30),CIE(30)
      DIMENSION C1E(30),C1U(30),C1(11)
C
      NC3=NC*3
      N1=N+1
      N2=N*2
C
      DO 1 I=1,NC
      DO 1 J=1,NC
    1 COV(NC+I,2*NC+J)=0.
      DO 2 J=1,NC
      DO 3 I=1,J
      COV(NC+I,NC+J)=0.
    3 COV(NC*2+I,NC*2+J)=0.
      COV(NC+J,NC+J)=VU
    2 COV(2*NC+J,2*NC+J)=VE
      DO 5 I=1,NC
      DO 5 J=1,I
      COV(I,NC+J)=0.
    5 COV(I,2*NC+J)=0.
```

```
      C
      C     COMPUTE RYU AND RYE.
      C
            DO 6 I=1,N
            CIU(I)=B(N+1-I)
            CIE(I)=C(N+1-I)
          6 AI(I+1)=A(N+1-I)
            AI(1)=0.
            AI(N+2)=1.
            CIE(N+1)=1.
            BI(1)=1.
            DO 8 K=1,NC
            NCU=N+K-2
            NCE=N+K-1
            IF(K.EQ.1) GO TO 7
            DO 17 I=1,N
            CIU(NCU+2-I)=CIU(NCU+1-I)
         17 CIE(NCE+2-I)=CIE(NCE+1-I)
            CIU(K-1)=0.
            CIE(K)=C(N)
            CIE(K-1)=0.
          7 RYU(K)=VU*CINT(AI,CIU,BI,N1,NCU,0)
          8 RYE(K)=VE*CINT(AI,CIE,BI,N1,NCE,0)
      C
      C     COMPUTE RYYU AND RYYE.
      C
            DO 4 I=1,N
          4 BI(I+1)=A(I)
            BI(1)=1.
            DO 9 I=1,N
          9 AI(I+1)=A(N+1-I)
            AI(1)=0.
            AI(N+2)=1.
            DO 10 I=1,N
            C1U(I)=B(N+1-I)
            C1E(I)=C(N+1-I)
         10 C1(I+1)=C(I)
            C1(1)=1.
            C1U(N1)=0.
            C1E(N1)=1.
            NX=N2+1
            CALL PMPY(CIU,NCU,B,N,C1U,N1)
            NS=NCU-1
            DO 18 I=1,NS
         18 CIU(NCU+1-I)=CIU(NCU-I)
            CIU(1)=0.
            CALL PMPY(CIE,NCE,C1,N1,C1E,N1)
      C
            NCU=NCU-2
            NCE=NCE-2
            DO 12 K=1,NC
            IF(K.EQ.1) GO TO 13
            KK=K-1
            DO 11 I=1,N2
            KI=KK+N2+1-I
            CIU(KI)=CIU(KI-1)
         11 CIE(KI+1)=CIE(KI)
            CIU(KK)=0.
            CIE(KK+1)=CIE(KK)
            CIE(KK)=0.
         13 NCU=NCU+1
            NCE=NCE+1
            RYYU(K)=VU*CINT(AI,CIU,BI,N1,NCU,N)
         12 RYYE(K)=VE*CINT(AI,CIE,BI,N1,NCE,N)
```

```
C
C
C     STORE RESULT IN COV.
C
      DO 14 J=1,NC
      DO 14 I=1,J
      COV(I,NC+J)=RYU(J-I+1)
      COV(I,2*NC+J)=RYE(J-I+1)
   14 COV(I,J)=RYYE(J-I+1)+RYYU(J-I+1)
      DO 15 I=1,NC3
      DO 15 J=1,I
   15 COV(I,J)=COV(J,I)
C
      RETURN
      END
```

```
      SUBROUTINE PMPY(Z,IDIMZ,X,IDIMX,Y,IDIMY)
C
C     POLYNOMIAL MULTIPLICATION
C     GIVEN
C     X,IDIMX,Y,IDIMY
C     X(T)=X(1)*T**(IDIMX-1)+...+X(IDIMX)
C     Y(T)=Y(1)*T**(IDIMY-1)+...+Y(IDIMY)
C     COMPUTED
C     Z,IDIMZ
C     Z(T)=X(T)*Y(T)
C     IDIMZ=IDIMX+IDIMY-1
C     Z(T)=Z(1)*T**(IDIMZ-1)+...+Z(IDIMZ)
C
C     AUTHORS IBM+I GUSTAVSSON+E BURSTRÖM +T SÖDERSTRÖM, ANNO DAZUMAL
C     SUBROUTINE REQUIRED
C            NONE
C
      DIMENSION Z(1),X(1),Y(1)
      IF(IDIMX*IDIMY) 10,10,20
   10 IDIMZ=0
      GO TO 50
   20 IDIMZ=IDIMX+IDIMY-1
      DO 30 I=1,IDIMZ
   30 Z(I)=0.
      DO 40 I=1,IDIMX
      DO 40 J=1,IDIMY
      K=I+J-1
   40 Z(K)=X(I)*Y(J)+Z(K)
   50 RETURN
      END
```

```
       SYSTEM
       ******

       NUMBER OF A-PARAMETERS     1
       NUMBER OF B-PARAMETERS     1
       NUMBER OF C-PARAMETERS     1
       NUMBER OF SAMPLING INTERVALS      500

       PARAMETERS
 -.80000   1.00000     .70000

       VARIANCE OF THE INPUT SIGNAL    1.0000
       VARIANCE OF THE NOISE           1.0000



       MODEL
       *****

       NUMBER OF A-PARAMETERS     2
       NUMBER OF B-PARAMETERS     2

       THEORETICAL LS ESTIMATE
       -----------------------

       PARAMETER ESTIMATES
-1.2857      .42249      1.0000      -.48573

       STANDARD DEVIATIONS
 .39949-01   .37905-01   .47958-01   .62417-01

       COVARIANCE MATRIX
 .15959-02  -.13682-02   .00000      .15959-02
-.13862-02   .14368-02   .00000     -.13682-02
 .00000      .00000      .23000-02   .00000
 .15959-02  -.13682-02   .00000      .38959-02

       LOSS FUNCTION
 575.00

       STANDARD DEVIATION OF THE NOISE
 1.0724
```

SYSTEM
******

NUMBER OF A-PARAMETERS 2
NUMBER OF B-PARAMETERS 2
NUMBER OF C-PARAMETERS 0
NUMBER OF SAMPLING INTERVALS 500

PARAMETERS

-.10000	-.56000	1.00000	.70000

VARIANCE OF THE INPUT SIGNAL 1.0000
VARIANCE OF THE NOISE 1.0000

MODEL
*****

NUMBER OF A-PARAMETERS 2
NUMBER OF B-PARAMETERS 2

THEORETICAL LS ESTIMATE

PARAMETER ESTIMATES

-.10000	-.56000	1.0000	.70000

STANDARD DEVIATIONS

.33507-01	.29368-01	.44721-01	.55882-01

COVARIANCE MATRIX

.11227-02	-.66922-03	.00000	.11227-02
-.66922-03	.86249-03	.00000	-.66922-03
.00000	.00000	.20000-02	.00000
.11227-02	-.66922-03	.00000	.31227-02

LOSS FUNCTION
500.00

STANDARD DEVIATION OF THE NOISE
1.0000